```
> restart:
```

Hornig.mws

SPIN AND TORSION FOR HORNIG's EXAMPLE

```
>  with(liesymm):with(linalg):with(plots):setup(x,y,z,t,s);
Warning, new definition for close
Warning, new definition for norm
Warning, new definition for trace
```

$$[x, y, z, t, s]$$

```
> defform(x=0,y=0,z=0,t=0,s=0,a=const,b=const,c=const,k=const,mu=const,omega=const
,m=const);
```

$$\mathrm{defform}(x = 0, y = 0, z = 0, t = 0, s = 0, a = const, b = const, c = const, k = const, \mu = const, \omega = const, m = const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [\mathrm{d}(x), \mathrm{d}(y), \mathrm{d}(z), \mathrm{d}(t)]$$

```
>
>
```

A choice for the 4 potentials to replicate Hornig's Fields

```
> A1:=0;A2:=0;A3:=(y^2/2-x^2);A4:=-z;
```

$$A1 := 0$$

$$A2 := 0$$

$$A3 := \frac{1}{2} y^2 - x^2$$

$$A4 := -z$$

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(s);
>
```

$$Action := \left(\frac{1}{2} y^2 - x^2\right) \mathrm{d}(z) + z\, \mathrm{d}(s)$$

```
> F:=wcollect(d(Action));
```

$$F := y\, ((\mathrm{d}(y))\ \&\wedge\ (\mathrm{d}(z))) - 2\, x\, ((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(z))) + ((\mathrm{d}(z))\ \&\wedge\ (\mathrm{d}(s)))$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

The three components of the Vector potential are:

```
> A:=[A1,A2,A3];
```

$$A := \left[0, 0, \frac{1}{2} y^2 - x^2\right]$$

The three components of the Magnetic field are:

```
> B:=(curl(A,[x,y,z])):B1:=factor(B[1]);B2:=factor(B[2]);B3:=factor(B[3]);
```

$$B1 := y$$

$$B2 := 2\, x$$

$$B3 := 0$$

Note that B3 vanishes , and the B field is of the Hedgehog type.

The three components of the Electric Field are

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);
```

$$E1 := 0$$

$$E2 := 0$$

$E3 := 1$

Topological Parity 4 form (Second Poncare invariant)

```
> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E):BdotB:=innerprod(B,B):ExB:=crossprod(E,B):
```

$EdotB := 0$

The Torsion current.

```
> ExA:=crossprod(E,A);Bphi:=[B1*A4,B2*A4,B3*A4];
```

$ExA := [0, 0, 0]$

$Bphi := [-y\,z, -2\,x\,z, 0]$

```
> TORS:=evalm(ExA+A4*B);
```

$TORS := [-y\,z, -2\,x\,z, 0]$

```
> AdotB:=factor(inner(A,B));
```

$AdotB := 0$

The helicity density vanishes!!!

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$TORSION := [-y\,z, -2\,x\,z, 0, 0]$

Divergence of the Torsion current.

```
> P2:=factor(diverge(TORSION,[x,y,z,t]));
```

$P2 := 0$

```
>
```

Now compute the current charge densities assuming B = uH and D=eE

```
> J:= evalm(curl(B,[x,y,z])/mu-epsilon*[diff(E1,t),diff(E2,t),diff(E3,t)]):
> J1:=factor(J[1]);J2:=factor(J[2]);J3:=factor(J[3]);rho:=factor(diverge(E,[x,y,z]));TAYLOR:=crossprod(J,B):CURLT:=curl(TAYLOR,[x,y,z]):factor(CURLT[1]);factor(diverge(TAYLOR,[x,y,z])):
>
```

$J1 := 0$

$J2 := 0$

$$J3 := \frac{1}{\mu}$$

$\rho := 0$

0

J is sigma*E but J is not rho*V as rho=0!

The Lorentz Force is not zero, but indeed is a gradient field. Note that the Lorentz force would induce a rotation if their was any free charge. Strange

```
>  Fl:=(evalm(rho*E+crossprod(J,B)));curl(Fl,[x,y,z]);
```

$$Fl := \left[-2\,\frac{x}{\mu}, \frac{y}{\mu}, 0\right]$$

$[0, 0, 0]$

```
> grad(y^2/2-x^2,[x,y,z])/mu;
```

$$\frac{[-2\,x, y, 0]}{\mu}$$

```
>
```

The SPin components are not zero and are not conserved.

```
> SSPIN:=evalm(crossprod(A,B/mu)+A4*epsilon*E);S4:=innerprod(A,epsilon*E);
>
```

$$SSPIN := \left[-2\frac{\left(\frac{1}{2}y^2 - x^2\right)x}{\mu}, \frac{\left(\frac{1}{2}y^2 - x^2\right)y}{\mu}, -z\,\varepsilon\right]$$

$$S4 := -\frac{1}{2}(-y^2 + 2\,x^2)\,\varepsilon$$

```
> SPIN:=[SSPIN[1],SSPIN[2],SSPIN[3],S4];
```

$$SPIN := \left[-2\frac{\left(\frac{1}{2}y^2 - x^2\right)x}{\mu}, \frac{\left(\frac{1}{2}y^2 - x^2\right)y}{\mu}, -z\,\varepsilon, -\frac{1}{2}(-y^2 + 2\,x^2)\,\varepsilon\right]$$

```
> P1:=factor(diverge(SPIN,[x,y,z,t]));
```

$$P1 := -\frac{1}{2}\frac{-10\,x^2 - y^2 + 2\,\varepsilon\,\mu}{\mu}$$

SO the SPin 3-form is not zero, and not conserved in your model field.
but the Torsion 3-form is not zero and conserved.
THe Lorentz force is proportional to a gradient..

```
>
```

Lets generalize the format.

You can write any function of Theta on the next line.

```
> Theta:=(alpha*y^2+beta*x^2);
>
```

$$\Theta := \alpha\,y^2 + \beta\,x^2$$

Define the four components of the 4-potentials here.

```
> A1:=0;A2:=0;A3:=(Theta);A4:=-z;
```

$$A1 := 0$$

$$A2 := 0$$

$$A3 := \alpha\,y^2 + \beta\,x^2$$

$$A4 := -z$$

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(s);
>
```

$$Action := (\alpha\,y^2 + \beta\,x^2)\,\mathrm{d}(z) + z\,\mathrm{d}(s)$$

```
> F:=wcollect(d(Action));
```

$$F := 2\,\alpha\,y\,((\mathrm{d}(y))\ \&\wedge\ (\mathrm{d}(z))) + 2\,\beta\,x\,((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(z))) + ((\mathrm{d}(z))\ \&\wedge\ (\mathrm{d}(s)))$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

IN ENGINEERING LANGUAGE:
The three components of the Vector potential are:

```
> A:=evalm([A1,A2,A3]);
```

$$A := [0, 0, \alpha\,y^2 + \beta\,x^2]$$

The three components of the Magnetic field are:

```
> B:=(curl(A,[x,y,z])):B1:=factor(B[1]);B2:=factor(B[2]);B3:=factor(B[3]);
```

$$B1 := 2\,\alpha\,y$$

$$B2 := -2\,\beta\,x$$

$$B3 := 0$$

Note that B3 vanishes, and the B field topology depends upon the anisotropy factors, alpha and beta.

The three components of the Electric Field are

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
  E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);
```

$$E1 := 0$$

$$E2 := 0$$

$$E3 := 1$$

Topological Parity 4 form (Second Poncare invariant)

```
> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E):BdotB:=innerprod(B,B):ExB:=c
  rossprod(E,B):
```

$$EdotB := 0$$

The spatial components of the Torsion current.

```
> ExA:=crossprod(E,A);Bphi:=[B1*A4,B2*A4,B3*A4];
```

$$ExA := [0, 0, 0]$$

$$Bphi := [-2\, \alpha\, y\, z,\, 2\, \beta\, x\, z,\, 0]$$

Note that ExA is zero, hence the spatial projection of the
"lines of T" are proportional to the "lines of B"

```
> TORS:=evalm(ExA+A4*B);
```

$$TORS := [-2\, \alpha\, y\, z,\, 2\, \beta\, x\, z,\, 0]$$

```
> AdotB:=factor(inner(A,B));
```

$$AdotB := 0$$

The Helicity density of this field is ZERO!

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$$TORSION := [-2\, \alpha\, y\, z,\, 2\, \beta\, x\, z,\, 0,\, 0]$$

Divergence of the Torsion current.

```
> P2:=factor(diverge(TORSION,[x,y,z,t]));
```

$$P2 := 0$$

The second Poincare invariant vanishes so the Helicity integral is conserved!
EVEN though AdotB = 0!!!

Now compute the currents densities

```
> J:= evalm(curl(B,[x,y,z])/mu-epsilon*[diff(E1,t),diff(E2,t),diff(E3,t)]):
> J1:=factor(J[1]);J2:=factor(J[2]);J3:=factor(J[3]);rho:=factor(diverge(E,[x,y,z]
  ));
>
>
```

$$J1 := 0$$

$$J2 := 0$$

$$J3 := -2\, \frac{\beta + \alpha}{\mu}$$

$$\rho := 0$$

J is proportional to E (an ohmic like result) , but rho is zero, so J cannot be rho*V!

NOte the special case, beta= - alpha. Then there is NO current density! The question arises as to what causes the Hedgehog B field. In your example, there is a finite current, due to the Anistropy, alpha <>beta. Interesting!

Compute the Lorentz force density FL, is it proportional to a Gradient,   Yes, if Curl(FL)=0

```
> FL:=(evalm(rho*E+crossprod(J,B))):FL1:=factor(FL[1]);FL2:=factor(FL[2]);FL3:=factor(FL[3]);CURLFL:=curl(FL,[x,y,z]);
```

$$FL1 := -4\,\frac{(\beta+\alpha)\,\beta\,x}{\mu}$$

$$FL2 := -4\,\frac{(\beta+\alpha)\,\alpha\,y}{\mu}$$

$$FL3 := 0$$

$$CURLFL := [0, 0, 0]$$

Compare to grad(Theat):

```
> grad(Theta,[x,y,z])/mu;
```

$$\frac{[2\,\beta\,x, 2\,\alpha\,y, 0]}{\mu}$$

Now compute the Spin components

```
>
> SSPIN:=evalm(crossprod(A,B/mu)+A4*epsilon*E);S4:=innerprod(A,epsilon*E);
>
```

$$SSPIN := \left[2\,\frac{(\alpha\,y^2+\beta\,x^2)\,\beta\,x}{\mu}, 2\,\frac{(\alpha\,y^2+\beta\,x^2)\,\alpha\,y}{\mu}, -z\,\varepsilon\right]$$

$$S4 := (\alpha\,y^2+\beta\,x^2)\,\varepsilon$$

```
> SPIN:=[SSPIN[1],SSPIN[2],SSPIN[3],S4];
```

$$SPIN := \left[2\,\frac{(\alpha\,y^2+\beta\,x^2)\,\beta\,x}{\mu}, 2\,\frac{(\alpha\,y^2+\beta\,x^2)\,\alpha\,y}{\mu}, -z\,\varepsilon, (\alpha\,y^2+\beta\,x^2)\,\varepsilon\right]$$

```
> P1:=factor(diverge(SPIN,[x,y,z,t]));
```

$$P1 := -\frac{-6\,\beta^2\,x^2-2\,\beta\,\alpha\,y^2-6\,\alpha^2\,y^2-2\,\alpha\,\beta\,x^2+\varepsilon\,\mu}{\mu}$$

```
>
```

The Spin components exist, and are not zero.
The First Poincare invariant is not zero

```
>
> INTERACTION:=(innerprod(J,A)-rho*A4);
```

$$INTERACTION := -2\,\frac{(\beta+\alpha)\,(\alpha\,y^2+\beta\,x^2)}{\mu}$$

```
> LAGRANGE:=factor(innerprod(B,B)/mu-innerprod(E,E)*epsilon);
```

$$LAGRANGE := -\frac{-4\,\alpha^2\,y^2-4\,\beta^2\,x^2+\varepsilon\,\mu}{\mu}$$

```
>
```